Making Mathematics Fun, Accessible, and Challenging using *Mathematica* 6

Paul C. Abbott
paul@physics.uwa.edu.au
School of Physics, M013
The University of Western Australia
Crawley WA 6009
AUSTRALIA

Abstract: The theme of this year's conference is "Making Mathematics Fun, Accessible and Challenging through Technology". Version 6 of *Mathematica*, released in May 2007, is instantly interactive, allowing teachers and students to easily construct live interfaces to illustrate and investigate mathematical concepts at all levels [1]. Over 2000 examples of these interactive capabilities, including a number written by high school students, are freely available at the Demonstrations Project website [2]. Other new features of Version 6, relevant to this conference, include high-quality adaptive visualization of functions and load-on-demand data of mathematical properties such as graphs, knots, lattices, and polyhedra.

1. Introduction

Version 1 of *Mathematica* was introduced in 1998 as a "System for doing Mathematics by Computer", combining numerics, symbolic computation, and graphics [3]. Almost 20 years later, *Mathematica* 6 was released, now consisting of seven tightly integrated components:



Figure 1 Overview of *Mathematica* 6.

Although there are a huge number of new features in *Mathematica* 6, from the perspective of ATCM, the most important ones are:

1. dramatic improvements in visualization and graphics;
2. sources for large volumes of data in hundreds of formats, including load-on-demand data of mathematical properties such as graphs, knots, lattices, and polyhedra;
3. a new fully integrated symbolic-dynamic interface for creation of complete dynamic interfaces;
4. the Demonstrations Project website [2].

This paper is itself a *Mathematica* Notebook [4] and includes many interactive examples. Unfortunately, there is no easy way to transfer this interactivity to paper. However, even if you do not have access to *Mathematica*, you can run the interactive examples using the freely available *Mathematica Player* [5].

2. Visualization and Graphics

Here are a few examples of the enhanced visualization and graphics.

Use `RegionPlot` to display the region in which $\left(x+\frac{1}{2}\right)^2+y^2\leq 1 \,\underline{\bigvee}\, \left(x-\frac{1}{2}\right)^2+y^2\leq 1$.

```
RegionPlot[(x + 1/2)^2 + y^2 ≤ 1 ⊻ (x - 1/2)^2 + y^2 ≤ 1, {x, -1.6, 1.6}, {y, -1, 1}]
```



Plot $\sin(x+\cos(y))$ over $[-\pi, 2\pi]\times[-\pi, 2\pi]$ restricted such that $x<y^2$ using `RegionFunction`.

```
Plot3D[Sin[x + Cos[y]], {x, -π, 2 π}, {y, -π, 2 π},
 RegionFunction → Function[{x, y}, x < y^2], Filling → Bottom, Mesh → 8]
```



Produce a 3D Contour plot of $x^2 + y^2 - z^2 = 0.1$.

```
ContourPlot3D[x² + y² - z² = 0.1, {x, -1, 1}, {y, -1, 1},
 {z, -1, 1}, ContourStyle → {{Yellow, Thickness[0.04]}}, Mesh → None]
```



Smooth shading is generated via geometric or analytic vertex normal computation.

```
ContourPlot3D[16 x³ + 16 y³ - 31 z³ + 24 x² z - 48 x² y - 48 x y² + 24 y² z - 54 √3 z² - 72 z = 1,
 {x, -2.5, 2.5}, {y, -2.5, 2.5}, {z, -2.5, 2.5}, Mesh → None, PlotPoints → 50,
 ContourStyle → {{Brown, Specularity[White, 20]}}, Background → Black,
 Boxed → False, Axes → False, SphericalRegion → True, Mesh → All]
```



Here is an example of an interesting algebraic surface.

```
ContourPlot3D[x^4 - 5 x^2 + y^4 - 5 y^2 + z^4 - 5 z^2 == -12.5, {x, -3, 3}, {y, -3, 3}, {z, -2.3, 2.3},
 PlotPoints → 50, MeshFunctions → {Function[{x, y, z}, x y z]}, MeshStyle → Purple,
 MeshShading → {Blue, Yellow, Green}, ContourStyle → Green, Boxed → False, Axes → False]
```



3. Data Sources

Next I present a few examples of the variety of data sources available in *Mathematica* 6.

Generate an array of Cayley graphs.

```
GraphicsGrid[Partition[GraphData /@ GraphData["CayleyGraph"], 14]]
```



If you take a polyhedron, cut along certain edges, and lay the whole thing flat the result is called a polyhedron net.

Here is a table of net images, indexed using `Tooltip`.

```
GraphicsGrid[Partition[
  (Graphics[Tooltip[PolyhedronData[#1, "NetImage"][[1]], PolyhedronData[#1, "Name"]]] &) /@
   PolyhedronData["NetImage", "Defined"], 15]]
```



Look-up those nuclear isotopes with mass-number A in the range $50 \le A \le 51$ and then draw a decay network.

```
is = Select[IsotopeData[], 50 ≤ IsotopeData[#, "MassNumber"] ≤ 51 &]
```

{Chlorine50, Chlorine51, Argon50, Argon51, Potassium50, Potassium51, Calcium50, Calcium51, Scandium50, Scandium51, Titanium50, Titanium51, Vanadium50, Vanadium51, Chromium50, Chromium51, Manganese50, Manganese51, Iron50, Iron51, Cobalt50, Cobalt51, Nickel50, Nickel51}

```
LayeredGraphPlot[
 Table[(IsotopeData[#, "Symbol"] & /@ (i → #)) & /@ IsotopeData[i, "DaughterNuclides"],
   {i, is}] // Flatten, DirectedEdges → True, VertexLabeling → True]
```



Visualize the caffeine molecule in 3D.

```
ChemicalData["Caffeine", "MoleculePlot"]
```



Generate a graphic of solar system orbit paths.

```
Graphics3D[{Red, AbsoluteThickness[1],
  AstronomicalData[#, "OrbitPath"] & /@ AstronomicalData["Planet"]}]
```



Generate a world map using a Mollweide projection.

```
CountryData["World", {"Shape", "Mollweide"}]
```

Since ATCM 2007 is held in Taiwan, find the country's population and show the shape of its electrical plug.

```
CountryData["Taiwan", #] & /@ {"Population", "ElectricalGridPlugImages"}
```

{23036087, {}}

4. Dynamic Interactivity

Version 6 includes a range of capabilities that define a new kind of dynamic interactive computing.

4.1 Manipulate

`Manipulate` is a new function that enables users to create dynamic user interfaces as easily as they might create a table or a plot. `Manipulate` is flexible and is tightly integrated with *Mathematica* notebooks, typesetting, and graphics.

Show the factors of $1 - x^n$ for integer parameter $1 \leq n \leq 120$.

```
Manipulate[Factor[x^n - 1], {n, 1, 120, 1}]
```



Animate a three-dimensional plot of $\sin(x\,y + t)$ over $[0, 4] \times [0, 4]$ as t varies from 0 to 2π.

```
Manipulate[Plot3D[Sin[x y + t], {x, 0, 4}, {y, 0, 4}, ImageSize → 200],
 {t, 0, 2 π}, ControlPlacement → Left, ControlType → VerticalSlider]
```



4.2 LocatorPane and ClickPane

Here are two examples of interactive graphic solutions to matrix differential equations of the form

$$\dot{\boldsymbol{x}}(t) = A \cdot \boldsymbol{x}(t), \tag{1}$$

for `Dynamic` values of the initial condition, $\boldsymbol{x}(0)$, provided by the location of a mouse click.

Visualize solutions using `LocatorPane`.

```
DynamicModule[{pt = {{4, 0}, {0, 6}, {-5, -5}}}, LocatorPane[Dynamic[pt],
  Dynamic@ParametricPlot[Evaluate[MatrixExp[( -1.1  0.9 ) t, #] & /@ pt],
                                            ( -1.4  0.3 )
    {t, 0, 10}, PlotRange → 7, ImageSize → 200]]]
```



Keep track of all solutions as you go using `ClickPane`.

```
DynamicModule[{s = {}},
 ClickPane[Framed@Dynamic@ParametricPlot[s, {t, 0, 10}, PlotRange → 5, ImageSize → 200],
  (AppendTo[s, MatrixExp[( -1.1  0.9 ) t].#]) &]]
                         ( -1.4  0.3 )
```



5. Teaching Examples

I have just started using this new interactive functionality in my teaching and would like to share with you some examples.

5.1 Complex Numbers Applied to Electric Circuits

Using complex numbers one can generalize Ohm's law so as to handle alternating current (AC) circuits containing resistors R, inductors L, and capacitors, C. First one needs to make clear the cartesian and polar representations of complex numbers.

The imaginary axis is perpendicular to the real axis. The point $\{x, y\} \equiv r\{\cos(\theta), \sin(\theta)\}$ is represented by $z = x + i\,y = r\,e^{i\theta}$ and is indicated by a `Locator`.



Now consider the circuit in Figure 2.



Figure 2 Series *LCR* circuit.

The complex impedance of this circuit is

$$Z = Z_L + Z_C + Z_R = R + \frac{1}{i\,\omega\,C} + i\,\omega\,L = R + i\left(\omega\,L - \frac{1}{\omega\,C}\right), \tag{2}$$

where $\omega = 2\,\pi\,\nu$ is the angular frequency of oscillation of the power source. Usually we are only interested in the magnitude and phase of $Z = |Z|\,e^{i\,\phi}$:

$$|Z| = \sqrt{R^2 + \left(\omega\,L - \frac{1}{\omega\,C}\right)^2}, \quad \tan(\phi) = \frac{y}{x} = \frac{\omega\,L - \frac{1}{\omega\,C}}{R}. \tag{3}$$

The average power $\langle P\rangle$ supplied by the power source in Figure 2 is

$$\langle P \rangle = \tilde{I}^2 R = \frac{\tilde{V}^2}{Z^2} R = \frac{\tilde{V}^2 R \omega^2}{\omega^2 R^2 + L^2 \left(\omega^2 - \omega_0^2\right)^2} = \frac{\tilde{V}^2}{R} \Bigg/ \left(1 + \frac{L^2}{R^2} \omega^2 \left(1 - \frac{\omega_0^2}{\omega^2}\right)^2\right), \tag{4}$$

where $\omega_0 = 1 \Big/ \sqrt{L C}$ is the **resonant frequency**. Plots of $\langle P \rangle$ versus ω are called **resonance curves**. Here is a demonstration that shows how resonant curves (4) depend upon the resistance R and the inductance L.

Log-linear plot of $\langle P \rangle$ versus ω for the series LCR circuit in Figure 2. The line marked with $\Delta\omega = R / L$ joins the two **half-power points** and is called the **full-width at half maximum** (FWHM). The (dimensionless) **quality factor** is $Q = \omega_0 / \Delta\omega$.



Once the behavior of simple circuits is understood, one can move to more complicated circuits such as that shown in Figure 3, where we measure the output voltage $\tilde{V}_{\text{out}}$ across R_2.



Figure 3 Filter circuit.

The total impedance of this circuit is complicated. However, we can effectively consider it as being built from two filter circuits as that shown in Figure 4.



Figure 4 Simplified filter circuit.

Considering the output, $\tilde{V}_{out}$, of the first filter as the input, $\tilde{V}_{in}$, to the second filter, one sees that the transfer function of this combination is, approximately, just the *product* of the transfer function of each filter:

$$\left|\frac{\tilde{V}_{out}}{\tilde{V}_{in}}\right| \approx \frac{1}{\sqrt{1+(\omega/\omega_1)^2}}\,\frac{1}{\sqrt{1+(\omega_2/\omega)^2}}, \text{ where } \omega_i = \frac{1}{R_i\,C_i}. \tag{5}$$

Log-linear plot of the *approximate* transfer function $|\tilde{V}_{out}/\tilde{V}_{in}|$ versus ω for a band-pass filter. The **centre frequency** is the **geometric mean** of the two half-power frequencies, $\omega_C = \sqrt{\omega_1\,\omega_2} \approx 316\,\text{rad}\cdot s^{-1}$.



5.2 Forced Oscillations and Resonance

Consider horizontal motion on a frictionless surface of a mass-spring system. The spring is loosely wound and can be compressed as well as stretched with force $F = -k\,x$, where k is the spring constant and x is the displacement from the equilibrium position.

Graphs of x (—), v (—), and a (—) for a mass-spring system.



The acceleration is always toward the equilibrium position ($x = 0$) and proportional to distance from the that point.

Now introduce a damping force, $F_d = -b\,v$, that opposes the motion and is proportional to the instantaneous velocity. Students should have an *intuitive feel* for the behaviour of this system; if the spring is stretched and then released, the mass will undergo damped simple harmonic motion with the amplitude reducing over time. And if the damping is sufficiently large, the mass will not oscillate at all.

From Newton's second law, the differential equation describing the motion of the mass, m, is

$$m\,\frac{d^2 x(t)}{dt^2} + b\,\frac{d x(t)}{dt} + k\,x(t) = 0, \tag{6}$$

and students should therefore have a ‘feel’ for the behavior of solutions to this differential equation, namely exponential decay of harmonic motion, that is

$$x(t) = x_0\, e^{-t/(2\tau)} \cos(\alpha\, t), \tag{7}$$

where $x_0 = x(0)$ is determined by the initial condition and, to satisfy the differential equation, the parameters are $\tau = 1/\Delta\omega = m/b$ and $\alpha = \omega_0 \sqrt{1 - 1/(4\,Q^2)}$, where $\omega_0 = \sqrt{k/m}$ is the **resonant frequency** and the (dimensionless) **quality factor** is $Q = \omega_0/\Delta\omega$.

An interesting question is what happens to the total energy. Energy is dissipated only when the mass is *moving*. For this reason, the total energy decays exponentially only on *average*.

The top plot displays the solution to the differential equation of the form $x_0\, e^{-t/(2\tau)} \cos(\alpha\, t)$. The **envelope** of the oscillation (—) is $\pm x_0\, e^{-t/(2\tau)}$. The bottom plot shows the kinetic energy (—), $K = \frac{1}{2}\, m\, v^2$, spring potential energy (—), $U = \frac{1}{2}\, k\, x^2$, and total energy (—), $E = K + U$, for the damped oscillator.



For an oscillatory driving force the differential equation reads

$$m\,\frac{d^2 x(t)}{dt^2} + b\,\frac{d x(t)}{dt} + k\,x(t) = F_0 \cos(\omega\, t), \tag{8}$$

where ω is the angular frequency of the driving force. By analogy with the series *LCR* circuit, the amplitude of a forced mechanical oscillator is

$$\Rightarrow \frac{x_0}{F_0} = \frac{1}{k}\,\frac{1}{\sqrt{\left(1 - \frac{\omega^2}{\omega_0^2}\right)^2 + \left(\frac{\omega}{Q\,\omega_0}\right)^2}} = \frac{1}{k}\, G(\omega), \tag{9}$$

where $G(\omega)$ is the (dimensionless) **transfer function**.

Plot of $G(\omega)$ as a function of ω for various values of Q. On resonance $G(\omega_0) = Q$.



At low frequency the response, $x_0 = F_0 / k$, depends only on the spring. On resonance the response depends on the Q. At high frequency the response is $x_0 \to F_0 / k\, (\omega_0 / \omega)^2 = F_0 \big/ \left(m\, \omega^2\right) \to 0$, which is the same as for a free mass.

5.3 Solitons

The Korteweg de Vries (KdV) equation, describing water waves in shallow channels, is given by

$$\frac{\partial u}{\partial t} - 6\, u\, \frac{\partial u}{\partial x} + \frac{\partial^3 u}{\partial x^3} = 0, \tag{10}$$

where $u(x, t)$ is the wave amplitude.

The initial value of the n-soliton solution is $u_n(x, 0) = -n(n+1)\,\mathrm{sech}^2(x) = 2\,\mathcal{V}_n(x)$. If $\mathcal{V}_n(x)$ is interpreted as the *potential* of the Schrödinger equation, that is

$$\frac{-1}{2}\, \frac{\partial^2 \psi\,(x)}{\partial x^2} + \mathcal{V}_n(x)\, \psi(x) = \mathcal{E}\, \psi(x), \tag{11}$$

where

$$\mathcal{V}_{\mathrm{n\_}}(\mathrm{x\_}) = -\frac{n\,(n+1)}{2}\, \mathrm{sech}^2(x);$$

the (unnormalized) solution is

$$\psi_{\mathrm{n\_,m\_}}(\mathrm{x\_}) = P_n^m(\tanh(x));$$

where $P_n^m(z)$ is the associated Legendre function, and the energy is

$$\mathcal{E}_{\mathrm{m\_}} = \frac{-m^2}{2};$$

For each n, there exists n bound states with $1 \le m \le n$. First we compute and save the normalization constants $\mathcal{N}_{n,m}$ determined by requiring $\mathcal{N}_{n,m}^2 \int_{-\infty}^{\infty} \psi_{n,m}(x)^2\, dx = 1$.

$$\mathcal{N} /: \mathcal{N}_{\mathrm{n\_,m\_}} := \mathcal{N} /: \mathcal{N}_{n,m} = 1 \Big/ \sqrt{\int_{-\infty}^{\infty} \psi_{n,m}(x)^2\, dx}$$

On one plot, display the potential $V_n(x)$ together with the normalized eigenfunctions, shifted downwards by the associated energy, $\mathcal{N}_{n,m}^2\, \psi_{n,m}(x)^2 + \mathcal{E}_m$, for $1 \le n \le 5$ and $1 \le m \le n$.

Manipulate$\left[\textbf{Plot}\left[\textbf{Evaluate}\left[\textbf{Join}\left[\textbf{Table}\left[\mathcal{N}_{n,m}^2\, \psi_{n,m}(x)^2 + \mathcal{E}_m, \{m, n\}\right], \{\mathcal{V}_n(x)\}\right]\right], \{x, -5, 5\}, \textbf{Filling} \to \textbf{Table}[m \to \mathcal{E}_m, \{m, n\}]\right], \{n, \textbf{Range}[5]\}, \textbf{SaveDefinitions} \to \textbf{True}\right]$



Here we visualize the time evolution of the "two-soliton" solution to the KdV equation (10). It is clear that there are now two waves; one slow and one fast.

$$\textbf{Manipulate}\left[\textbf{Plot}\left[-\frac{12\,(4\cosh(8\,t - 2\,x) + \cosh(64\,t - 4\,x) + 3)}{(3\cosh(28\,t - x) + \cosh(36\,t - 3\,x))^2}, \{x, -40, 40\}, \textbf{PlotRange} \to \{-8.1, 0.1\}\right], \{t, -2, 2, 0.25\}\right]$$



5.4 Kepler Equation

The *Kepler equation*, $s = u + e \sin(s)$ — critical in celestial mechanics — relates the mean anomaly u (a parameterization of time) of a body in an elliptical orbit of eccentricity e to the body's eccentric anomaly s (a parameterization of polar angle) [6]. Computing s is a commonly-used intermediate step to the calculation of planetary position as a function of time.

Consider the following related geometrical example: Pick an arbitrary point F inside the unit circle centered at O. Let P be the point on the circle closest to F and pick another point Q elsewhere on the circle. The location of F and Q are dynamic and can be modified by moving the `Locator` associated with each point. Define s and e as pictured below and let u be twice the area of the shaded sector PFQ.

Observe that $\frac{1}{2}u = \text{Area(sector } P0Q) - \text{Area}(\Delta F0Q) = \frac{1}{2}s - \frac{1}{2}e\sin(s)$.



6. Demonstrations

Here are a few more advanced examples of the functionality of `Manipulate` freely available for download from the Demonstrations Project website [2].

Produce a random set of up to 200 points and join each point to a specified number of nearest neighbors, computed using the `Nearest` function [7].



Here is a very nice demonstration of least-squares curve fitting [8].



The following problem was presented at the International Mathematical Olympiad (IMO) of 2006 in Slovenia [9]: Let ABC be a triangle with incentre I. A point P in the interior of the triangle satisfies ∠PBA + ∠PCA = ∠PBC + ∠PCB. Show that AP ≥ AI, and that equality holds if and only if $P = I$.

Observe that P moves along the brown circle with center at the intersection of the circumcircle and the bisector of the angle A.



7. Conclusion

In this talk I demonstrated several of the new capabilities in *Mathematica* Version 6, showing how much fun one can have when the technology is high-level, robust, portable, and easy to use.

References

[1] http://www.wolfram.com

[2] http://demonstrations.wolfram.com

[3] http://www.wolfram.com/products/mathematica/history.html

[4] Available at http://physics.uwa.edu.au/pub/ATCM/2007 as a *Mathematica* Notebook

[5] http://www.wolfram.com/products/player/

[6] http://mathworld.wolfram.com/KeplersEquation.html

[7] http://demonstrations.wolfram.com/NearestNeighborNetworks/

[8] http://demonstrations.wolfram.com/LinearAndQuadraticCurveFittingPractice/

[9] http://demonstrations.wolfram.com/AnIMOTriangleProblem/